
ABSTRACT  
The ODS EXCEL destination has made sharing SAS® reports and graphs much easier. What is even 
more exciting is that this destination is available for use regardless of the platform. This is extremely 
useful when reporting is performed on remote servers. This presentation goes through the basics of using 
the ODS EXCEL destination and shows specific examples of how to use this in a remote environment. 
Examples for both SAS® on Windows and in SAS® Enterprise Guide® are provided. 

INTRODUCTION  
What is ODS Excel, and why use it? 

ODS Excel creates xlsx files in native format, regardless of the operating system being used.   Other 
methods of getting SAS information included ODS HTML, Proc Export, and DDE (among others). 

ODS HTML created files that could be opened in Excel, but usually didn’t produce desired look, were not 
native Excel output, and created very large files.  PROC EXPORT can export data, but not graphs and 
report results.  The formatting is unreliable as well.  DDE is flexible, but can be quirky and it may not be 
supported in the future.  As with proc export, it outputs data, not reports and graphs, and in some cases 
data must be manipulated considerably before output. 

Using ODS Excel, you can put your output into Excel, keeping the formatting you desire.  You can add 
graphics as well.  And it’s just plain cool to do something new! 

GETTING STARTED WITH ODS EXCEL 
As with other ODS destinations, the ODS Excel “sandwich” has a top and bottom layer.  An example of 
the top layer is below, and includes some of the available options: 

ods listing close; /* (note that this isn’t necessary, but prevents duplicate output)*/
ods excel file = "&my_excel_file." 
   options (embed_footnotes_once = 'on' 
      embedded_footnotes='on' 
      embed_titles_once = 'on' 
      embedded_titles='on' 
      gridlines='off' 
      sheet_interval='proc' 
      sheet_name="Movement Matrix"); 

Additional information on the options available in ODS Excel can be found in the following locations: 

http://blogs.sas.com/content/sasdummy/?s=ods+excel 
http://support.sas.com/rnd/base/ods/ODS_Excel_Dest_tips.pdf 

AFTER REPORT CODE: 
ods excel close; 
ods listing; 

The Excel file is created when the “ods excel close; command is executed.   
The “ods listing” command turns the regular output back on (this is needed only if it was turned off in the 
step above). 

http://blogs.sas.com/content/sasdummy/?s=ods+excel
http://blogs.sas.com/content/sasdummy/?s=ods+excel
http://support.sas.com/rnd/base/ods/ODS_Excel_Dest_tips.pdf

MOVING THE FILE: 
In most cases, we’re working in a UNIX/LINUX environment through PC SAS using rsubmit, or via 
Enterprise Guide. 
 
Multiple choices are available for moving the Excel file into a Windows environment, depending on the 
SAS tool being utilized. 
 
The following method works in PC SAS without the need for an additional step outside of SAS: 
 
rsubmit; 
proc download infile = "&my_excel_file."  
    outfile = "H:\My Documents\mm.xlsx" binary; 
run; 
endrsubmit; 
 
Please note the difference from normal download code – the “binary” option is needed.  Without it, the 
download will appear to work, but it will not be readable in Excel.   

DEMONSTRATION – REAL EXAMPLE 

The problem: 

We have an important reporting requirement that we produce officially once a year.   

• We report on loans and dollars by a number of categories.   

• When changes are made either based on changes in guidance or corrections, we need to 
understand the impact on our report 

• We created a movement matrix to show the movement of dollars and units from category to 
category (by year) and wanted to publish it in a readable format in Excel 

The example shown is for PC SAS, and thus has all of the rsubmit and endrsubmit commands in the 
appropriate places. 

For demonstration purposes, a simple random number generation was used: 

rsubmit; 
%let my_excel_file = ~/mm.xlsx;  
%sysrput my_excel_file = &my_excel_file.; 
 
 
/* Create user defined format which will be used to "bucket" the values */ 
proc format; 
   value bkt 
 low - <-200 = "Negative 3" 
 -200 - <-100 = "Negative 2" 
 -100 - <0    = "Negative 1" 
 0 - <100     = "Positive 1" 
 100 - <200   = "Positive 2" 
 200 - high   = "Positive 3"; 
run; 
 
/* create some data for testing */ 
data rand1; 
   do i = 1 to 100; 
 x1 = rand("Normal")*100; 
 xmt = ranuni(0)*100; 
 output; 

Note – use of tilde (~) here 
goes to user’s default 
location. Any location on 
the server can be used. 

   end; 
run; 
data rand2; 
   do i = 1 to 100; 
 x2 = rand("Normal")*100; 
 output; 
   end; 
run; 
data rand3; 
   do i = 1 to 100; 
 x3 = rand("Normal")*100; 
 output; 
   end; 
run; 
 
/* combine the data */ 
data randomdata; 
   merge rand1 rand2 rand3; 
   by i; 
   Bucket1 = put(x1,bkt.); 
   Bucket2 = put(x2,bkt.); 
   Bucket3 = put(x3,bkt.); 
run; 
endrsubmit; 

 

THE REPORT, INCLUDING THE “ODS SANDWICH”: 
/* close default listing output destination */ 
ods listing close; 
rsubmit; 
/* open the ods Excel destination - output from all code between this 
statement and the "ods excel close;" statement will be directed to 
&my_excel_file.*/ 
ods excel file = "&my_excel_file." 
   options (embed_footnotes_once = 'on' 
      embedded_footnotes='on' 
      embed_titles_once = 'on' 
      embedded_titles='on' 
      gridlines='off' 
      sheet_interval='proc' 
      sheet_name="Movement Matrix"); 
 
title1 'Normal Distribution - Random Numbers'; 
title2 'Movement Matrix'; 
title3 'Dollars'; 
proc tabulate data=randomdata 
   format=dollar24.;  
   where Bucket1 ne Bucket2; 
   var xmt; 
   class Bucket1 /order=unformatted missing; 
   class Bucket2 /order = unformatted missing; 
   table /* ROW Statement */ 
      Bucket1 ='First Range'*Bucket2 = "Second Range", 
        /* COLUMN Statement */ 
      xmt ='Dollars'* sum=" " /box = 'Movement Matrix First to Second' 
misstext = '$0'; 

 table /* ROW Statement */ 
      Bucket2 = "Second Range"*Bucket1 ='First Range', 
            /* COLUMN Statement */ 
      xmt ='Dollars'* sum=" " /box = 'Movement Matrix Second to First' 
misstext = '$0'; 
run; 
 
title3 'Count'; 
proc tabulate data=randomdata 
   format=comma7.;  
   where Bucket1 ne Bucket2; 
   var xmt; 
   class Bucket1 /order=unformatted missing; 
   class Bucket2 /order = unformatted missing; 
   table   /* ROW Statement */ 
      Bucket1 ='First Range'*Bucket2 = "Second Range", 
           /* COLUMN Statement */ 
      xmt ='Count'* n=" " /box = 'Movement Matrix First to Second' misstext = 
          $0'; 
   table  /* ROW Statement */ 
 Bucket2 = "Second Range"*Bucket1 ='First Range', 
  /* COLUMN Statement */ 
 xmt ='Count'* n=" " /box = 'Movement Matrix Second to First' misstext =  
         '$0'; 
run; 
title3 'Dollars'; 
proc tabulate data=randomdata 
   format=dollar24.;  
   where Bucket1 ne Bucket3; 
   var xmt; 
   class Bucket1 /order=unformatted missing; 
   class Bucket3 /order = unformatted missing; 
   table /* ROW Statement */ 
      Bucket1 ='First Range'*Bucket3 = "Third Range", 
   /* COLUMN Statement */ 
      xmt ='Dollars'* sum=" " /box = 'Movement Matrix First to Third'  
        misstext = '$0'; 
   table  /* ROW Statement */ 
      Bucket3 = "Third Range"*Bucket1 ='First Range', 
  /* COLUMN Statement */ 
      xmt ='Dollars'* sum=" " /box = 'Movement Matrix Third to First'  
       misstext = '$0'; 
run; 
 
title3 'Count'; 
proc tabulate data=randomdataformat=comma7.;  
   where Bucket1 ne Bucket3; 
   var xmt; 
   class Bucket1 /order=unformatted missing; 
   class Bucket3 /order = unformatted missing; 
   table /* ROW Statement */ 
      Bucket1 ='First Range'*Bucket3 = "Third Range", 
  /* COLUMN Statement */ 
      xmt ='Count'* n=" " /box = 'Movement Matrix First to Third' misstext =  
        '$0'; 
   table  /* ROW Statement */ 
      Bucket3 = "Third Range"*Bucket1 ='First Range', 

  /* COLUMN Statement */ 
      xmt ='Count'* n=" " /box = 'Movement Matrix Third to First' misstext =  
        '$0'; 
run; 
title 'Histogram of 3 Random Selections'; 
 
/* create a bar chart! */ 
ods graphics / height=400 width=800 noborder; 
proc sgplot data=randomdata; 
   histogram x1; 
   histogram x2; 
   histogram x3; 
run; 
 
/* close the ods excel file */ 
ods excel close; 
endrsubmit; 
ods listing; 
  

SAMPLE OUTPUT 
The code will create 5 tabs, with the last tab being the histogram.  The report in Excel will look like this: 

Movement Matrix First to 
Second 

Dollars 

First Range Second Range $251 
Negative 1 Negative 2  
 Positive 1 $410 
 Positive 2 $120 
 Positive 3 $12 
Negative 2 Negative 1 $203 
 Positive 1 $96 
 Positive 2 $65 
Negative 3 Negative 1 $76 
 Negative 2 $11 
 Positive 1 $86 
Positive 1 Negative 1 $602 
 Negative 2 $258 
 Negative 3 $45 
 Positive 2 $348 
 Positive 3 $131 
Positive 2 Negative 1 $304 
 Negative 2 $196 
 Positive 1 $183 
Positive 3 Negative 1 $17 
 Negative 2 $54 
 Positive 1 $70 
 

The last tab will contain a histogram looking something like this: 

PROGRAMMATICALLY DOWNLOADING FILE IN ENTERPRISE GUIDE: 
For users of Enterprise Guide, the method of downloading will be different, but can still be done 
programmatically: 

After running the program, double-click on the Process Flow in the Project Tree, or select View | Process 
Flow from the main menu 

• Select the xlsx file (it will have a plain white icon and if you wrote the file to your home directory 
path, you should see the start of the path /users/ap…) and Right-Click to bring up a menu 

• Select Export and then Export … as a Step In Project…  

• Click Next. 

• Select Local Computer and click the Browse button. 

• Navigate to the path where you want the file to land. 

• Change Files of type: to All File Types 

• Notice that the file name is the path and file name from the server, but all slashes and dots have 
been replaced with underscores, and a .txt has been added by default.  

• Change the file name to something appropriate and make sure that it ends with the proper suffix: 
.xlsx 

• Click Save. 

• Cllick Next. 

• Review the selections and if everything looks ok, click Finish 

 
FINALLY – WHAT TO DO WITH THE EXCEL FILE ON THE REMOTE SERVER? 

You most likely won’t be using the Excel file on your remote server.  And most companies would prefer if 
you kept space usage limited.  So it’s a good idea to clean up the files from the server once the job is 
complete and you’ve downloaded the Excel file to your computer or network. 

Several methods can be used: 

• Manually remove items using a telnet client 

• System commands executed from SAS 

• Simplest one for me is using an “x” command: 

rsubmit; 
x "rm &my_excel_file."  
endrsubmit; 

CONCLUSION 
Try it yourself!   

For much of my career, I have been providing reporting and data via SAS and placing the information into 
Excel for others’ use.  I have used a combination of PROC EXPORT and DDE (and occasionally the 
libname method) for getting the data into Excel and placing it where I want it.  For this reason, much of 
my programming has been done to develop summary datasets and manipulate the data for export.  I 
have spent little time in learning to use report procedures like PROC REPORT, PROC TABULATE, and 
various graph procedures.  The capabilities of placing report output directly into Excel will make me much 
more likely to want to learn more about using these direct reporting procedures. 

The ability to use this tool remotely makes it much more usable when working with large datasets.  In 
many cases it wouldn’t be feasible to run these reports on a local machine.  Look for opportunities to use 
this new tool in SAS 9.4. 

CONTACT INFORMATION  
Your comments and questions are valued and encouraged. Contact the author at: 

Tom Bugg  
Wells Fargo Home Mortgage 
303-953-1651 
thomas.b.bugg@wellsfargo.com 

“SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS 
Institute Inc. in the USA and other countries. ® indicates USA registration.   

